Math 1 Two hours test May 17, 2016.

JE 13.5.16/JKL April 28, 2017

Problem 1

```
> restart;
```

Given the function

```
> f:=x->sqrt(2*x-1);
```

$$f := x \to \sqrt{2x-1} \quad \textbf{(1.1)}$$

Question 1

The function is defined for $2x-1 \ge 0 \Leftrightarrow x \ge \frac{1}{2}$. I.e. $\mathrm{Dm}(f)$ is the interval $[\frac{1}{2};\infty[$.

Question 2

With the development point $x_0 = 1$ we get

```
> P3:=unapply(mtaylor(f(x),x=1,4),x);
```

$$P3 := x \to x - \frac{1}{2}(x-1)^2 + \frac{1}{2}(x-1)^3 \quad \textbf{(1.2.1)}$$

Question 3

```
> diff(f(x),x,x,x,x);
```

$$-\frac{15}{(2x-1)^{7/2}} \quad \textbf{(1.3.1)}$$

According toTaylors formula with $x_0 = 1$ there exists for every $x \ge \frac{1}{2}$ a ξ between x and 1, so

$f(x) = P_3(x) + R_3(x)$, where

$f(x) - P_3(x) = R_3(x)$

$$= \frac{f^{(4)}(\xi)}{4!}(x-1)^4 = -\frac{15}{4!\,(2\xi-1)^{7/2}}(x-1)^4 = -\frac{5}{8}\cdot\frac{1}{(2\xi-1)^{7/2}}(x-1)^4 .$$

For $x = \frac{3}{2}$ there exists a ξ between $\frac{3}{2}$ and 1, so

$$f\left(\frac{3}{2}\right) = P_3\left(\frac{3}{2}\right) - \frac{5}{8}\cdot\frac{1}{(2\xi-1)^{7/2}}\left(\frac{3}{2}-1\right)^4 = P_3\left(\frac{3}{2}\right) - \frac{5}{8}\cdot\frac{1}{(2\xi-1)^{7/2}}\left(\frac{1}{2}\right)^4 .$$

If we use $P_3\left(\frac{3}{2}\right)$ instead of $f\left(\frac{3}{2}\right)$, then the error is

$$\left|f\left(\frac{3}{2}\right) - P_3\left(\frac{3}{2}\right)\right| = \left|R_3\left(\frac{3}{2}\right)\right| = \frac{5}{8}\cdot\frac{1}{(2\xi-1)^{7/2}}\left(\frac{1}{2}\right)^4 \le \frac{5}{8\cdot 2^4} = \frac{5}{2^7} ,$$

since $1 \le \xi \le \frac{3}{2}$.

```
> P3(3/2);
```

$$\frac{23}{16} \quad \textbf{(1.3.2)}$$

```
> evalf(P3(3/2));
```

$$1.437500000 \quad (1.3.3)$$

```
> evalf(5/2^7);
```

$$0.03906250000 \quad (1.3.4)$$

```
> evalf(P3(3/2)-5/2^7);
```

$$1.398437500 \quad (1.3.5)$$

Addition to question 3:

Since $f\left(\frac{3}{2}\right) - P_3\left(\frac{3}{2}\right) = R_3\left(\frac{3}{2}\right) < 0$ we get

$P_3\left(\frac{3}{2}\right) - \frac{5}{2^7} \leq f\left(\frac{3}{2}\right) < P_3\left(\frac{3}{2}\right)$ and thus

$\frac{23}{16} - \frac{5}{2^7} \leq f\left(\frac{3}{2}\right) < \frac{23}{16}$.

Therefore : $1.3984 < f\left(\frac{3}{2}\right) = \sqrt{2} < 1.4375$.

Problem 2

```
> restart;with(LinearAlgebra):with(plots):
```

Given the symmetric matrix

```
> A:=<<288/25,84/25>|<84/25,337/25>>;
```

$$A := \begin{bmatrix} \frac{288}{25} & \frac{84}{25} \\ \frac{84}{25} & \frac{337}{25} \end{bmatrix} \quad (2.1)$$

Question 1

```
> g:=(x,y)->evalb(x[1]<y[1]):
> ev:=sort(Eigenvectors(A,output=list),g);
```

$$ev := \left[\left[9, 1, \left\{\begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix}\right\}\right], \left[16, 1, \left\{\begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix}\right\}\right]\right] \quad (2.1.1)$$

From this we read that $\mathbf{A} \in \mathbb{R}^{2 \times 2}$ has the eigenvalues 9 and 16 with $E_9 = \operatorname{span}\left\{\begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix}\right\}$ and

$E_{16} = \operatorname{span}\left\{\begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix}\right\}$.

Since **A** is symmetric, the two eigenvector spaces E_9 and E_{16} are orthogonal.

```
> v1:=ev[1,3,1];
```

$$v1 := \begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix} \quad (2.1.2)$$

is a basis for E_9 .

```
> q1:=(-1)*Normalize(v1,Euclidean);
```

$$q1 := \begin{bmatrix} \frac{4}{5} \\ -\frac{3}{5} \end{bmatrix} \quad \textbf{(2.1.3)}$$

is an orthonormal basis for E_9 .

```
> v2:=ev[2,3,1];
```

$$v2 := \begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix} \quad \textbf{(2.1.4)}$$

is a basis for E_{16} .

```
> q2:=Normalize(v2,Euclidean);
```

$$q2 := \begin{bmatrix} \frac{3}{5} \\ \frac{4}{5} \end{bmatrix} \quad \textbf{(2.1.5)}$$

where $\mathbf{q}_2 = \widehat{\mathbf{q}_1}$,is an ortonormal basis for E_{16} .

The eigenvectors $(\mathbf{q}_1, \mathbf{q}_2)$ are then an orthonormal basis for $\mathbb{R}^2$ equipped with the ordinary dot product.

If we put

```
> Q:=<q1|q2>;
```

$$Q := \begin{bmatrix} \frac{4}{5} & \frac{3}{5} \\ -\frac{3}{5} & \frac{4}{5} \end{bmatrix} \quad \textbf{(2.1.6)}$$

and

```
> Lambda:=DiagonalMatrix([9,16]);
```

$$\Lambda := \begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix} \quad \textbf{(2.1.7)}$$

then $\mathbf{Q}$ is positive orthogonal and

$$\mathbf{\Lambda} = \mathbf{Q}^{-1}\mathbf{A}\ \mathbf{Q} = \mathbf{Q}^{\mathrm{T}}\mathbf{A}\ \mathbf{Q}$$

Check using Maple:

```
> Transpose(Q).Q;Determinant(Q);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$1 \quad \textbf{(2.1.8)}$$

```
> Q^(-1).A.Q;
```

$$\begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix} \quad \textbf{(2.1.9)}$$

```
> Transpose(Q).A.Q;
```

$$\begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix} \qquad \textbf{(2.1.10)}$$

Question 2

An ellipse E in an ordinary orthogonal (x, y)-coordinate system in the plane is given by the matrix equation

$$\begin{bmatrix} x & y \end{bmatrix} \mathbf{A} \begin{bmatrix} x \\ y \end{bmatrix} = 144 .$$

From question 1 we now get that $(\mathrm{O} ; \mathbf{q}_1, \mathbf{q}_2)$ is a new ordinary orthogonal coordinate system in the plane emerging from a rotation of the given coordinate system about O.
The relation between the old coordinates (x, y) and the new coordinates (x_1, y_1) for a point P in the plane is given by

$$\begin{bmatrix} x \\ y \end{bmatrix} = \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} .$$

The equation of the ellipse in the new coordinates is

$$\begin{bmatrix} x_1 & y_1 \end{bmatrix} \mathbf{Q}^{\mathrm{T}} \mathbf{A}\, \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = 144 \Leftrightarrow \begin{bmatrix} x_1 & y_1 \end{bmatrix} \begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = 144 \Leftrightarrow 9\, x_1^{\;2} + 16\, y_1^{\;2} = 144 \Leftrightarrow$$

$$\frac{x_1^{\;2}}{16} + \frac{y_1^{\;2}}{9} = 1.$$

Rom this we read that the major semi-axis og the ellipse is a = 4 and the minor semi-axis is b = 3.

The X_1 -axis, that is the line through O with the direction vector $\mathbf{q}_1$, has the equation: $y = -\frac{3}{4} x$.

The Y_1 -axis, that is the line through O with the direction vector $\mathbf{q}_2$, has the equation $y = \frac{4}{3} x$.

```
> <x|y>.A.<x,y>;
```

$$\left(\frac{288}{25} x + \frac{84}{25} y\right) x + \left(\frac{84}{25} x + \frac{337}{25} y\right) y \qquad \textbf{(2.2.1)}$$

```
> E:=expand(%)=144;
```

$$E := \frac{288}{25} x^2 + \frac{168}{25} x\, y + \frac{337}{25} y^2 = 144 \qquad \textbf{(2.2.2)}$$

```
> implicitplot({E,y=-3/4*x,y=4/3*x},x=-5..5,y=-5..5,scaling=
  constrained);
```



Problem 3

```
> restart;with(LinearAlgebra):with(plots):
```

For a smooth function $f: \mathbb{R}^2 \rightarrow \mathbb{R}$ with $f(0,0) = 0$ a vector field $\mathbf{V}$ in the (x, y)-plane is given by $\mathbf{V}(x, y) = \nabla f(x, y) = (f'_x(\text{x,y}) \ , f'_y(\text{x,y})) = (x - y^2 + 1, -2\,xy)$

Question 1

$\nabla f(x, y) = (0, 0) \Leftrightarrow x = 0$ and $y = \pm 1$ or $y = 0$ and $x = -1$.
All stationary points for f are then $(0, 1)$, $(0, -1)$ and $(-1, 0)$.

Question 2

If f has a proper local maximum or a proper local minimum in a point, then the point must be a stationary point, since f has no exception points.

```
> diff(x-y^2+1,x);
```

$$1 \quad \textbf{(3.2.1)}$$

```
> diff(-2*x*y,y);
```

$$-2\,x \quad \textbf{(3.2.2)}$$

```
> diff(x-y^2+1,y);
```

$$-2\,y \quad \textbf{(3.2.3)}$$

The Hessian matrix for f in the point (x, y) is

```
> H(x,y):=<<1,-2*y>|<-2*y,-2*x>>;
```

$$H(x,y) := \begin{bmatrix} 1 & -2y \\ -2y & -2x \end{bmatrix} \tag{3.2.4}$$

```
> H(0,1):=subs(x=0,y=1,H(x,y));
```

$$H(0,1) := \begin{bmatrix} 1 & -2 \\ -2 & 0 \end{bmatrix} \tag{3.2.5}$$

```
> Eigenvalues(H(0,1),output=list);
```

$$\left[\frac{1}{2} + \frac{1}{2}\sqrt{17}, \frac{1}{2} - \frac{1}{2}\sqrt{17}\right] \tag{3.2.6}$$

Since the two eigenvalues for $\mathbf{H}(0,1)$ have opposit signs, f has neither a proper local maximum nor a proper local minimum in the stationary point $(0,1)$.

```
> H(0,-1):=subs(x=0,y=-1,H(x,y));
```

$$H(0,-1) := \begin{bmatrix} 1 & 2 \\ 2 & 0 \end{bmatrix} \tag{3.2.7}$$

```
> Eigenvalues(H(0,-1),output=list);
```

$$\left[\frac{1}{2} + \frac{1}{2}\sqrt{17}, \frac{1}{2} - \frac{1}{2}\sqrt{17}\right] \tag{3.2.8}$$

Since the two eigenvalues for $\mathbf{H}(0,-1)$ have opposit signs, f has neither a proper local maximum nor a proper local minimum in the stationary point $(0,-1)$.

$$\begin{bmatrix} 1 & 2 \\ 2 & 0 \end{bmatrix}$$

$$f \tag{3.2.9}$$

```
> H(-1,0):=subs(x=-1,y=0,H(x,y));
```

$$H(-1,0) := \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix} \tag{3.2.10}$$

```
> Eigenvalues(H(-1,0),output=list);
```

$$[2, 1] \tag{3.2.11}$$

Since both eigenvalues for $\mathbf{H}(-1,0)$ are positive,
f has a proper local minimum in the stationary point $(-1,0)$.
Therfore:
f has exactly one proper local minimum viz. in the point $(-1,0)$ and no proper local maxima.

Question 3

Stair line K: $(0,0) \rightarrow (x,0) \rightarrow (x,y)$.
Since the gradient field $\nabla f = (V_x, V_y)$ is given, we can find f with $f(0,0) = 0$ by the formula

$$f(x,y) = f(0,0) + \mathrm{Tan}(\nabla f, K) = \mathrm{Tan}(\nabla f, K) = \int_0^x V_x(t, 0)\mathrm{d}t + \int_0^y V_y(x,t)\mathrm{d}t$$

$$= \int_0^x (t+1)\mathrm{d}t + \int_0^y -2\,xt\,\mathrm{d}t = \frac{1}{2}x^2 + x - xy^2.$$

Question 4

```
> f:=(x,y)->1/2*x^2+x-x*y^2;
```

$$f := (x, y) \rightarrow \frac{1}{2} x^2 + x - x y^2 \quad \textbf{(3.4.1)}$$

```
> 'f(-1,0)'=f(-1,0);
```

$$f(-1, 0) = -\frac{1}{2} \quad \textbf{(3.4.2)}$$

```
> contourplot(f(x,y),x=-3..3,y=-2..2,contours=25);
```



Problem 4

```
> restart;with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
```

A = $\{(x, y) | 0 \leq \text{x} \leq 2 \text{ og } -\frac{\pi}{2} \leq y \leq \frac{\pi}{2}\}$.

We consider the function

```
> h:=(x,y)->x*cos(y);
```

$$h := (x, y) \rightarrow x \cos(y) \quad \textbf{(4.1)}$$

for $(x, y) \in$ A.

The graph surface $F = \{(x, y, z) \mid$

$$0 \le x \le 2, -\frac{\pi}{2} \le y \le \frac{\pi}{2}, z = h(x, y)\}.$$

```
> plot3d(h(x,y),x=0..2,y=-Pi/2..Pi/2,scaling=constrained,view=0.
  .2);
```



Question 1

The parametric representation for *F*

```
> r:=<u,v,h(u,v)>;
```

$$r := \begin{bmatrix} u \\ v \\ u\cos(v) \end{bmatrix} \qquad \textbf{(4.1.1)}$$

where $u \in [0; 2]$ and $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$.

```
> r1:=diff~(r,u);
```

$$r1 := \begin{bmatrix} 1 \\ 0 \\ \cos(v) \end{bmatrix} \qquad \textbf{(4.1.2)}$$

```
> r2:=diff~(r,v);
```

$$r2 := \begin{bmatrix} 0 \\ 1 \\ -u \sin(v) \end{bmatrix} \quad \textbf{(4.1.3)}$$

The normal vector of the surface is

```
> N:=kryds(r1,r2);
```

$$N := \begin{bmatrix} -\cos(v) \\ u \sin(v) \\ 1 \end{bmatrix} \quad \textbf{(4.1.4)}$$

Question 2

In this question and in the next question we furthermore consider a vector field **V** in the (x, y, z)-space which has the properties
$\text{Div}(\mathbf{V})(x, y, z) = x + y + z$ and $\mathbf{Curl}(\mathbf{V})(x, y, z) = (3z, 3x, 3y)$.
With the chosen orientation of the closed boundary curve ∂F for F we get
$\mathbf{n}_F = \dfrac{\mathbf{N}(u, v)}{|\mathbf{N}(u, v)|}$ (right-hand rule).
From Stokes' theorem we then get

$$\text{Cirk}(\mathbf{V}, \partial F) = \text{Flux}(\mathbf{Curl}(\mathbf{V}), F) = \int_F \mathbf{n}_F \cdot \mathbf{Curl}(\mathbf{V})\, d\mu = \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_0^2 \mathbf{N}(u, v) \cdot \mathbf{Curl}(\mathbf{V})(\mathbf{r}(u,v))\, du$$

dv

```
> integrand:=prik(N,<3*u*cos(v),3*u,3*v>);
```

$$integrand := -3\cos(v)^2 u + 3u^2 \sin(v) + 3v \quad \textbf{(4.2.1)}$$

```
> Int(Int(integrand,u=0..2),v=-Pi/2..Pi/2)=int(int(integrand,u=
  0..2),v=-Pi/2..Pi/2);
```

$$\int_{-\frac{1}{2}\pi}^{\frac{1}{2}\pi} \int_0^2 \left(-3\cos(v)^2 u + 3u^2 \sin(v) + 3v\right) du\, dv = -3\pi \quad \textbf{(4.2.2)}$$

Question 3

The parametric representation for the solid region

```
> R:=<u,v,w*u*cos(v)>;
```

$$R := \begin{bmatrix} u \\ v \\ w\, u \cos(v) \end{bmatrix} \quad \textbf{(4.3.1)}$$

where $u \in [0; 2]$, $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$ and $w \in [0; 1]$.

$\partial\Omega$ is the closed surface of Ω oriented with an outward pointing unit normal vector.
From Gauss' theorem we then get

$$\text{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \text{Div}\,(\mathbf{V})\,d\mu = \int_0^1 \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_0^2 \text{Div}\,(\mathbf{V})(\mathbf{R}(u,v,w))\,|JR(u,v,w)|\,du\,dv\,dw$$

```
> M:=<diff~(R,u)|diff~(R,v)|diff~(R,w)>;
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ w\cos(v) & -w\,u\sin(v) & u\cos(v) \end{bmatrix} \quad \textbf{(4.3.2)}$$

```
> JR:=Determinant(M);
```

$$JR := u\cos(v) \quad \textbf{(4.3.3)}$$

that is ≥ 0 , since $u \in [0; 2]$ and $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$.

```
> integrand:=expand((w*u*cos(v)+v+u)*JR);
```

$$integrand := u^2\cos(v)^2 w + u\cos(v)\,v + u^2\cos(v) \quad \textbf{(4.3.4)}$$

```
> Int(Int(Int(integrand,u=0..2),v=-Pi/2..Pi/2),w=0..1)=int(int
  (int(integrand,u=0..2),v=-Pi/2..Pi/2),w=0..1);
```

$$\int_0^1 \int_{-\frac{1}{2}\pi}^{\frac{1}{2}\pi} \int_0^2 \left(u^2\cos(v)^2 w + u\cos(v)\,v + u^2\cos(v)\right) du\,dv\,dw = \frac{2}{3}\pi + \frac{16}{3} \quad \textbf{(4.3.5)}$$